

Lesson 9: Ifs - More User Input

Lesson aim:

To learn about conditions. If *this* happens do *that*. This is an essential part of probably every computer program you've ever used.

Why:

Conditions are one of the most important parts of coding. Try and imagine all the if conditions for any website or app. For example, *if* you press enter in google *then* make the search.

{code blocks}: are sections of code that go between a "{" and a "}".

Lets introduce "if" statements, along with their **{code blocks}**, **mousePressed** and what they all do:

Example

```
void setup() {
  size(600,600); //← in setup code block
}

void draw() {
  background(255,0,0);
  if(mousePressed == true){
    background(0,255,0); //← this is all in the if code block
  }
}
```

Try it! What happens: _____

Note: in programming "==" means "is equal to". So:

`if(mousePressed == true)` means: if **mousePressed** is equal to true then do the following **{code block}**, otherwise don't.

Anatomy of an "if" - How To Use "if" Statements

This is how processing reads "if" **{code blocks}**:

```
if (what is in here is true) {
  do what is in the {code block} after the if,
}
otherwise just continue after the end of the code block.
```

That's it!

- If what is between the "(" and the ")" is **true**, the **DO** run the code in the **{code block}**.
- If what is between the "(" and the ")" is **false**, then **DO NOT** run the code in the **{code block}**.

`mousePressed` is one keyword, lets introduce another: `keyPressed`.

What do you think `keyPressed` does? _____

Use the reference guide to check if you are right! (see lesson 7 on, if you can't remember how to get to the reference guide)

Tasks:

1. Create a new program, start its name with "**Lesson9**".
2. Make that program draw a rectangle only if you press the mouse.
3. Add a circle that appears whether you press the mouse or not.
4. Extend that program so that if you press a key, you change the colour of the background.
5. Make a program that draws a very basic face - (use 3-6 more shapes), but make the eyes change colour if you press the mouse, and the rest of the face change colour if you press a key.

Buttons and keys

Now we know how to check if **any** button on the mouse is pressed and if **any** button on the keyboard is pressed. What if we want to check if the letter 'a' is pressed or the left mouse button is clicked?

We do this using the `key` and `mouseButton` checks. These cannot just be true or false though. Have a look at the below example.

```
void setup() {
  size(600, 600);
}

void draw() {
  background(255);
  fill(0);
  if (keyPressed) {
    if (key == 'a') {
      fill(0, 255, 0);
    }
  }
  if (mousePressed) {
    if (mouseButton == LEFT) {
      ellipse(300, 300, 500, 500);
    }
    if (mouseButton == RIGHT) {
      rect(50, 50, 500, 500);
    }
  }
}
```

Note: we need to check if any key is pressed, and in that code block check which specific key is pressed, and we need to check if any mouse button is pressed and in that block check which button.

How `mouseButton`, `mousePressed`, `key` and `keyPressed` work:

key words	Options (case sensitive)		
<code>mousePressed</code>	true		false
<code>mouseButton</code>	LEFT	CENTER	RIGHT
<code>keyPressed</code>	true		false
<code>key</code>	'a' to 'z'	'A' to 'Z'	'0' - '9'

Tasks:

- Save the example from above; we will be making changes to it.
- At the moment the colour changes when you press 'a' .
 - Add another key that changes colour to red. (Now we can change the shape to 3 different colours)
 - Add yet another key for the colour yellow (now we have 4 colour choices)
- Change the background, have two different keys: one to make the background white another to make it blue.